CIVIL ENGINEERS
POCKET COMPUTER MONTHLY

BEAMS - SIMPLE AND CONTINUOUS

In this issue we will examine beams and will provide PC-1/PC-2 programs for

1) fixed end moment
2) four-span moment distribution

Using the programs provided herein, we will design a continuous four span beam for three loading patterns.

PROGRAM - FIXED END MOMENT

Program solves for left and right fixed end moments for 1) Uniform, 2) Concentrated, and 3) Ramp (tapered) Loading Conditions.

1. Uniform Load



2. Concentrated Loads



3. Ramp (Slope-Down Left)



4. Ramp (Slope-Down Right)



FIXED END MOMENT

Program Listing/PC-1

```
2Ø  "A" CLEAR: BEEP 4: PAUSE
    "PRINT ALL FEMS?,OR SUM"
4Ø  INPUT "ALL OR SUM? (A/S)";V$
9Ø  INPUT "ENTER SPAN, FT=";L
12Ø INPUT "NO. OF UNIF LOADS=";N:
    IF N=0 THEN 27Ø
135 IF V$="A" PRINT "*FEM,UNIF.
    LOADS*":PRINT " "
14Ø FOR M=1 TO N
15Ø INPUT "ENTER DIST L(A), FT=";B:
    IF B>L THEN 15Ø
16Ø INPUT "ENTER DIST L(B),FT=";C:
    IF C>L THEN 16Ø
17Ø INPUT "ENTER UNIT LOAD,
    K/FT=";W
18Ø A=(B+C)/2/L:GOSUB 8ØØ:Q=YY/12:
    R=Ø:GOSUB 86Ø
23Ø IF V$="S" THEN 245
24Ø GOSUB 91Ø:GOTO 25Ø
245 GOSUB 89Ø
25Ø NEXT M
27Ø INPUT "NO. OF CONC. LOADS=";N:
    IF N=Ø THEN 38Ø
285 IF V$="A" PRINT "*FEM, CONC.
    LOADS*":PRINT " "
29Ø FOR M=1 TO N
3ØØ INPUT "ENTER DIST L(A),FT=";B:
    IF B>L THEN 3ØØ
31Ø INPUT "ENTER CONC. LOAD,
    KIPS=";P
32Ø S=L-B:E=BSSP/L/L:F=BBSP/L/L
33Ø IF V$="S" THEN 35Ø
34Ø GOSUB 91Ø:GOTO 36Ø
35Ø GOSUB 89Ø
36Ø NEXT M
38Ø INPUT "NO. OF RAMP (SDL)
    LOADS=";N
39Ø IF N=Ø THEN 53Ø
395 IF V$="A" PRINT "FEM,RAMP
    (SDL) LDS":PRINT " "
4ØØ FOR M=1 TO N
41Ø INPUT "ENTER DIST L(A),FT=";
    B:IF B>L THEN 41Ø
```

```
420 INPUT "ENTER DIST L(B),FT=";C:
    IF C>L THEN 420
430 INPUT "ENTER MAX. UNIT
    LD.,K/FT=";W
440 A=((C-B)*2/3+B)/L:GOSUB 800:
    GOSUB 830
470 GOSUB 860:E=E/2:F=F/2:IF
    V$="S" THEN 505
500 GOSUB 910:GOTO 510
505 GOSUB 890
510 NEXT M
530 INPUT "NO. OF RAMP(SDR)
    LOADS=";N
540 IF N=0 THEN 655
545 IF V$="A" PRINT "FEM,RAMP(SDR)
    LDS":PRINT " "
550 FOR M=1 TO N
560 INPUT "ENTER DIST. L(A),FT=";B:
    IF B>L THEN 560
570 INPUT "ENTER DIST. L(B),FT=";C:
    IF C> L THEN 570
580 INPUT "ENTER MAX. UNIT
    LD., K/FT=";W
590 A=((C-B)*2/3+L-C)/L:GOSUB
    800: GOSUB 830:GOSUB 860:I=E:E=F/
    2:F=I/2
630 IF V$="S" THEN 645
640 GOSUB 910:GOTO 650
645 GOSUB 890
650 NEXT M
655 IF V$="A" THEN 690
660 PRINT "****FEM(SUM)****":
    PRINT " "
670 PRINT USING "########.##";
    "FEM,LEFT=":PRINT -T
680 PRINT "FEM,RGHT=":PRINT
    U:PRINT " "
690 PRINT " ":PRINT " ":PRINT
    " ":END
800 D=1-A:X=1-3*D:G=1-3*A:Y=
    (C-B)/L:RETURN
830 Q=YY/18:R=-YYY/135
840 RETURN
860 E=(QX+ADD+R)*WYLL
865 F=(QG+AAD-R)*WYLL
870 RETURN
890 T=T+E:U=U+F
900 RETURN
910 PRINT USING "########.##";
    "FEM,LEFT=":PRINT -E
920 PRINT "FEM,RGHT=":PRINT
    F:PRINT " ":RETURN
```

FIXED END MOMENT

Program Listing/PC-2

```
10  "A" CLEAR:WAIT 100
20  BEEP 4: PRINT "PRINT ALL
    FEMS?, OR"
30  PRINT "FEM (SUM) ONLY?"
40  INPUT "ALL OR SUM? ENTER
    (A/S)";V$
60  PRINT "THIS PGM ANALYZES FOR"
70  PRINT "LOADS:UNIF;CONC;RAMP"
90  INPUT "ENTER SPAN, FT= ";L
110 PRINT "FEM FOR UNIF LOADS"
120 INPUT "NO. OF UNIF. LOADS= ";N
130 IF N=0 THEN 260
135 IF V$="A" LPRINT "**FEM,
    UNIF. LOADS**":LPRINT
140 FOR M=1 TO N
150 INPUT "ENTER DIST L(A),FT= ";B:
    IF B>L THEN 150
160 INPUT "ENTER DIST L(B),FT= ";C:
    IF C>L THEN 160
170 INPUT "ENTER UNIT LOAD,
    K/FT= ";W
180 A=(B+C)/2/L:GOSUB 800
190 Q=Y^2/12:R=0
210 GOSUB 860
230 IF V$="S" THEN 245
240 GOSUB 1000:GOTO 250
245 GOSUB 890
250 NEXT M
260 PRINT "FEM FOR CONC LOADS"
270 INPUT "NO. OF CONC. LOADS= ";N
280 IF N=0 THEN 370
285 IF V$="A" LPRINT "**FEM,
    CONC. LOADS**":LPRINT
290 FOR M=1 TO N
300 INPUT "ENTER DIST L(A),FT= ";B:
    IF B>L THEN 300
310 INPUT "ENTER CONC. LOAD,
    KIPS= ";P
320 S=L-B:E=B*S^2/L^2*P:F=B^2
    *S/L^2*P
330 IF V$="S" THEN 350
340 GOSUB 1000:GOTO 360
350 GOSUB 890
360 NEXT M
370 PRINT "FEM FOR RAMP LOADS (SDL)"
380 INPUT "NO. OF RAMP (SDL)
    LOADS= ";N
390 IF N=0 THEN 520
395 IF V$="A" LPRINT "FEM,RAMP
    (SDL) LOADS":LPRINT
400 FOR M=1 TO N
```

```
41Ø  INPUT "ENTER DIST L(A),
     FT= ";B: IF B>L THEN 41Ø
42Ø  INPUT "ENTER DIST L(B),FT= ";C:
     IF C>L THEN 42Ø
43Ø  INPUT "ENTER MAX. UNIT
     LD., K/FT= ";W
44Ø  A=((C-B)*2/3+B)/L:GOSUB 8ØØ
45Ø  GOSUB 83Ø
47Ø  GOSUB 86Ø:E=E/2:F=F/2
49Ø  IF V$="S" THEN 5Ø5
5ØØ  GOSUB 1ØØØ:GOTO 51Ø
5Ø5  GOSUB 89Ø
51Ø  NEXT M
52Ø  PRINT "FEM FOR RAMP LOADS
     (SDR)"
53Ø  INPUT "NO. OF RAMP (SDR)
     LOADS = ";N
54Ø  IF N=Ø THEN 66Ø
545  IF V$="A" LPRINT "FEM, RAMP
     (SDR) LOADS":LPRINT
55Ø  FOR M=1 TO N
56Ø  INPUT "ENTER DIST L(A),
     FT= ";B:IF B>L THEN 56Ø
57Ø  INPUT "ENTER DIST L(B),
     FT= ";C:IF C>L THEN 57Ø
58Ø  INPUT "ENTER MAX. UNIT
     LD., K/FT= ";W
59Ø  A=((C-B)*2/3+L-C)/L:GOSUB 8ØØ
6ØØ  GOSUB 83Ø
61Ø  GOSUB 86Ø:I=E:E=F/2:F=I/2
63Ø  IF V$="S" THEN 645
64Ø  GOSUB 1ØØØ:GOTO 65Ø
645  GOSUB 89Ø
65Ø  NEXT M
66Ø  IF V$="S" LPRINT:LPRINT
     "****SUM OF FEM****":LPRINT
67Ø  IF V$="S" BEEP 4:LPRINT USING
     "########.##";"FEM,LEFT=":
     -INT (1ØØ*T+.5)/1ØØ
68Ø  IF V$="S" LPRINT "FEM,RGHT= ";
     INT (1ØØ*U+.5)/1ØØ:LPRINT
69Ø  LF 3:END
8ØØ  D=1-A:X=1-3*D:G=1-3*A:Y=
     (C-B)/L:RETURN
83Ø  Q=Y^2/18:R=-Y^3/135
84Ø  RETURN
86Ø  E=(Q*X+A*D^2+R)*W*Y*L^2
865  F=(Q*G+A^2*D-R)*W*Y*L^2
87Ø  RETURN
89Ø  T=T+E:U=U+F
9ØØ  RETURN
1ØØØ BEEP 4:LPRINT USING
     "########.##";"FEM,LEFT=";
     -INT (1ØØ*E+.5)/1ØØ
1010 LPRINT "FEM,RGHT=";INT
     (1ØØ*F+.5)/1ØØ:LPRINT:RETURN
```

PROGRAM - FOUR-SPAN MOMENT DISTRIBUTION

Program solves for end moments in up to four spans.

Program Listing/PC-2

For Uniform Loading Only

Note: The symbol, @, used in this program, is obtained by pressing SHIFT =.

```
 1Ø  "A" CLEAR:DIM A(5Ø)
 4Ø  INPUT "I(4)= ",U,V,W,X
 5Ø  INPUT "SPANS (4)= ",E,F,G,H
 6Ø  INPUT "UNIF. LDS(4)= ",A,B,C,D
 65  INPUT "NO. OF CYCLES= ";P
 7Ø  I=U/E/(U/E+V/F):J=1-I
 8Ø  K=V/F/(V/F+W/G):L=1-K
 9Ø  M=W/G/(W/G+X/H):N=1-M
1ØØ  Y=27:FOR Q=1 TO 4:GOSUB
     5ØØ:A(Y)=-T:Y=Y+1:A(Y)=T:
     Y=Y+1:NEXT Q
18Ø  FOR Q=27 TO 34:A(Q+8)=A(Q):NEXT Q
21Ø  FOR Q=1 TO P
22Ø  FOR Y=44 TO 48 STEP 2:A(Y)=
     -(A(Y-8)+A(Y-7))*@(Y-35):
     A(Y+1)=-(A(Y-8)+A(Y-7))*@(Y-34)
225  NEXT Y
23Ø  A(43)=-A(35):A(5Ø)=-A(42):A(27)=A(35)
     +A(43)
24Ø  FOR Y=28 TO 32 STEP 2:A(Y)=A(Y)+
     A(Y+16)+A(Y+15)/2
25Ø  A(Y+1)=A(Y+1)+A(Y+17)+A(Y+18)/2:
     NEXT Y
26Ø  A(34)=A(42)+A(5Ø)
27Ø  FOR Y=35 TO 41 STEP 2:A(Y)=A(Y+
     9)/2:A(Y+1)=A(Y+8)/2:NEXT Y
34Ø  NEXT Q
35Ø  FOR Q=28 TO 33:A(Q)=A(Q)-A(Q+8):
     NEXT Q
36Ø  USING:Y=1:LPRINT "END MOMENTS
     (L/R)":LPRINT:FOR Q=27 TO
     33 STEP 2:LPRINT "SPAN # ";Y
365  USING"########.###":LPRINT
     A(Q)
37Ø  LPRINT A(Q+1):LPRINT:USING:
     Y=Y+1:NEXT Q
41Ø  LF 3:END
5ØØ  R=@(Q+4):S=@(Q)
51Ø  T=S*R^2/12:RETURN
```

For PC-1 Program Listing, change as follows:

```
 1Ø  "A" CLEAR
22Ø  FOR Y=44 TO 48 STEP 2: A(Y)=
     -(A(Y-8)+A(Y-7))*A(Y-35):A(Y+1)=
     -(A(Y-8)+A(Y-7))*A(Y-34)
36Ø  USING:Y=1:PRINT "END MOMENTS
     (L/R)":PRINT " ":FOR Q=27
     TO 33 STEP 2:PRINT "SPAN# ";Y
365  USING "#######.###":PRINT A(Q)
37Ø  PRINT A(Q+1):PRINT " ":USING:
     Y=Y+1:NEXT Q
41Ø  PRINT " ":PRINT " ":PRINT
     " ":END
5ØØ  R=A(Q+4):S=A(Q)
```

The following version of the Four-Span Moment Distribution Program solves for end moments based on Fixed End Moments which have been calculated by the user. These Fixed End Moments are input when called for by the program.

For PC-2 Program Listing, change as follows:

```
 6Ø  Omit this line
1ØØ  WAIT 6Ø:Y=1:FOR Q=27 TO
     33 STEP 2:PRINT "FEM/L,SPAN#
     ";Y:INPUT A(Q)
```

Add Line 11Ø.

```
11Ø  PRINT "FEM/R,SPAN# ";Y:INPUT
     A(Q+1):Y=Y+1:NEXT Q
```

Omit Lines 5ØØ and 51Ø

For PC-1 Program Listing, change as follows:

```
 1Ø  "A" CLEAR
 6Ø  Omit this line
1ØØ  Y=1:FOR Q=27 TO 33 STEP
     2:PAUSE "FEM/L,SPAN# ";Y:INPUT
     A(Q)
```

Add Line 11Ø

```
11Ø  PAUSE "FEM/R,SPAN# ";Y:INPUT
     A(Q+1):Y=Y+1:NEXT Q
22Ø  FOR Y=44 TO 48 STEP 2:A(Y)=
     -(A(Y-8)+A(Y-7))*A(Y-35):
     A(Y+1)=-(A(Y-8)+A(Y-7))* A(Y-34)
36Ø  USING:Y=1:PRINT "END MOMENTS
     (L/R)":PRINT " ":FOR Q=27
     TO 33 STEP 2:PRINT "SPAN
     # ";Y
365  USING "#######.###":PRINT
     A(Q)
37Ø  PRINT A(Q+1):PRINT " ":
     USING":Y=Y+1:NEXT Q
41Ø  PRINT " ":PRINT " ":PRINT
     " ":END
```

Omit Lines 5ØØ and 51Ø

We offer Cassette Tapes with documentation at $20. each for:

1) Statistics-Confidence Program, PC-1 or PC-2
2) General Computation Program with or without printer, PC-1 or PC-2
3) Wood Column; Wood Foundation Wall; Wood Members - Combined Axial and Bending - PC-1 or PC-2
4) Steel Column and Steel Beam-Column, PC-2
5) Hazen-Williams and Chezy Manning Flow Formulas, PC-1 or PC-2
6) Walls Retaining Drained Sand, Active Case, PC-1 or PC-2
7) Pratt Truss, PC-2 (requires 8K Module)
8) Beams - Fixed End Moment and Four Span Moment Distribution (two versions), PC-1 or PC-2
9) Accounts Receivable, PC-2. This program prints detailed invoices, including aging. Prepares summary of business during the time period. Suitable for up to five hourly categories of charge rates. Requires 8K module.

Note: Civil Engineers Pocket Computer Monthly supports Radio Shack's PC-1 and PC-2 (Sharp PC-1211 and PC-1500). We believe our software will be helpful to civil engineers who have other equipment.

The software provided in this issue is solely for educational and experimental purposes. It is supplied "as-is" without warranty of any kind. We do not assume any liability for any direct, indirect, incidental or consequential damages relating to the use or application of the programs or information contained herein.

WORKED OUT EXAMPLES

We will demonstrate the use of the Four Span Moment Distribution Program for 4, 3, and 2 span problems, Uniform Load. We will use the version of the Moment Distribution Program which requires input of fixed end moments because it is more "general"; thus, it should more fully illustrate the way in which the program works.

The user must select the number of cycles of moment distribution desired. When performing Moment Distribution by hand methods, we have often stopped at three or four cycles on the basis that "three significant figures," or sliderule accuracy, was thereby achieved. Since six or eight is as easy to run on the computer, as three or four, we now tend toward a larger number of cycles. The program running time is increased, however. For PC-1, running time may seem extraordinarily long.

The program will accept variable span lengths, and variable moments of inertia - but constant within each span.

Fixed end moments may be calculated using the FEM program provided herewith. The user is warned to pay careful attention to signs.

A few example problems have been worked out by AISC; see their "Manual of Steel Construction", Pages 2-124 to 2-127, Eighth Edition.

Problem #1

See Example #39 Pg. 2-127. Continuous Beam - Four Equal Spans - All Spans Uniformly Loaded over the Entire Span.

The Moment Distribution program asks for four values of Moment of Inertia, I. Input four equal positive non-zero values, say, 1. Then the program asks for four values of span. Use 10 ft. for all four spans. The program now asks for number of cycles; use six. For FEM values use

FEM/L = -100/12 = -8.33333
FEM/R = 100/12 = 8.33333

The sign convention used for input of fixed end moments is "clockwise positive."

```
END MOMENTS(L/R).

SPAN#  1
                0.000
               10.709

SPAN#  2
              -10.709
                7.161

SPAN#  3
               -7.161
               10.709

SPAN#  4
              -10.709
                0.000
```

Note that the values provided by the AISC Manual are:

0; -10.71; -7.14; -10.71; 0

which is at slight variance with the computer output. (The signs, as used in the AISC Manual, are those which are conventional for moment diagrams, i.e., a moment which produces "tension top" is negative.)

The same problem is now rerun using eight cycles instead of six.

```
END MOMENTS(L/R).

SPAN#  1
                0.000
               10.713

SPAN#  2
              -10.713
                7.145

SPAN#  3
               -7.145
               10.713

SPAN#  4
              -10.713
                0.000
```

Note that the output now agrees with the AISC values.

Problem #2

See Example #37, Pg. 2-127. Continuous Beam - Four Equal Spans - First, Second and Fourth Spans Uniformly Loaded.

Input the same values when asked, except that, for the third span, FEM/L=FEM/R=Ø.

```
END MOMENTS(L/R).

SPAN#  1
              0.000
             12.052

SPAN#  2
            -12.052
              1.788

SPAN#  3
             -1.788
              5.802

SPAN#  4
             -5.802
              0.000
```

After eight cycles, the computer output matches the AISC values: Ø; -12.Ø5; -1.79; -5.8Ø; Ø

Problem #3

See Example #35, Pg. 2-126. Continuous Beam - Three Equal Spans - First and Third Spans Uniformly Loaded.

Treat the fourth as a "false" span.

Input I values as follows: 1, 1, 1, 1E-1Ø

Input Span values as follows: 1Ø, 1Ø, 1Ø, 1Ø. A non-zero value is required for all spans, to avoid ERROR 38 (division by zero).

Input FEM values as follows:

-1ØØ/12 ; 1ØØ/12
Ø ; Ø
-1ØØ/12 ; 1ØØ/12
Ø ; Ø

Using eight cycles, the output is as follows:

```
END MOMENTS(L/R).

SPAN#  1
              0.000
              5.000

SPAN#  2
             -5.000
              5.000

SPAN#  3
             -5.000
              0.000

SPAN#  4
              0.000
              0.000
```

Thus, the computer output agrees with the AISC values:

Ø; -5.Ø; -5.Ø; Ø

Problem #4

See Example #29, Pg. 2-124. Continuous Beam - Two Equal Spans - First Span Uniformly Loaded.

The third and fourth spans are treated as false spans:

Input I values: 1, 1, 1E-1Ø, 1E-1Ø

Input span values: 1Ø, 1Ø, 1Ø, 1Ø

Input FEM values: -1ØØ/12; 1ØØ/12
Ø; Ø
Ø; Ø
Ø; Ø

After eight cycles the output agrees with the AISC values: Ø; -6.25; Ø

```
END MOMENTS(L/R).

SPAN#  1
              0.000
              6.249

SPAN#  2
             -6.250
              0.000

SPAN#  3
              0.000
              0.000

SPAN#  4
              0.000
              0.000
```

Problem #5 See Errata May Pg. 5

Our final example will demonstrate the flexibility of the programs provided in this issue:

Continuous Beam-Four Non-Equal Spans - Four Loading Conditions including:

Span #1 Uniform Loading, Stepped



Span #2 Pattern of Concentrated Loads



Span #3 Ramp Loading, Partial Span, Slope Down Left



Span #4 Ramp Loading, Partial Span, Slope Down Right



The printed PC-1 output for FEM's is as follows:

FEM, Uniform Loads:

```
****FEM(SUM)****

FEM,LEFT=
              -72.56
FEM,RGHT=
               62.43
```

FEM. Concentrated Loads:

```
****FEM(SUM)****

FEM,LEFT=
              -76.25
FEM,RGHT=
              101.75
```

FEM, Ramp Load, SDL :

```
FEM,RAMP(SDL)LDS

FEM,LEFT=
              -50.85
FEM,RGHT=
               72.90
```

FEM, Ramp Load, SDR:

```
FEM,RAMP(SDR)LDS

FEM,LEFT=
              -33.07
FEM,RGHT=
               25.87
```

These values are input into the Moment Distribution Program:

Input I Values: 3Ø, 4Ø, 5Ø, 25

Input Span Lengths: 18, 2Ø, 25, 13

Input FEM Values:

-72.56 ; 62.43
-76.25 ; 101.75
-50.85 ; 72.90
-33.07 ; 25.87

After eight cycles the PC-2 end moments are as follows:

```
END MOMENTS(L/R)

SPAN#  1
                0.000
               94.161

SPAN#  2
              -94.161
               76.853

SPAN#  3
              -76.853
               52.818

SPAN#  4
              -52.818
                0.000
```

It may be necessary to repeatedly rerun the Moment Distribution program using revised I values until a beam has been successfully designed. Of course, if the beam is comprised of a single section throughout (no change in I), all I values may be taken as unity (1).

REVIEW OF SOME OF PC-2'S CAPABILITIES

FUNCTION KEYS

The six function keys incorporate three "reserve" areas (I, II, III) so that 18 functions are available; the function area which is currently operational is evidenced by the roman numeral shown in the display, which can be changed using the ▲▼ key.

The functions are defined by the user while in the "Reserve" mode. Press SHIFT MODE . Press F1. Using ◄ and ► keys place the cursor to the right of F1. Enter desired symbol, string, statement, command or function. Press ENTER . Continue until all function keys are defined. Then exit the reserve mode (Press MODE) The function keys are now available for use as defined. For example we have loaded Reserve Area I as follows:

F1: $
F2: "
F3: #
F4: :
F5: ,
F6: ;

We have found these to be efficient during programming work, saving one keystroke.

CHANGING BATTERIES

The black dot in the right side of the display will be "on" if good batteries are installed, and/or when the AC adaptor power supply is plugged in.

This black dot on the display is intended as a gauge on the condition of the batteries. If it fades batteries are weak or exhausted.

Since the PC-2 incorporates an automatic shutoff feature (while retaining memory), the batteries in the computer tend to last a long time; however, when considering replacing these four alkaline AA batteries, note that it is not good procedure to allow the batteries to become fully depleted, as the pre-programmed reserve functions, and all programs and data, may thereby be lost. During battery change, work with reasonable speed (no panic); with any luck all program/data/memory/ reserve functions will be retained intact. Note that your AC adaptor works in the printer-plotter, and in the computer. The computer (not the printer/plotter) could be connected to the AC adaptor auxiliary power supply while changing batteries. However we have never done so and have always retained the memory fully.

Note: Batteries in the computer are not recharged while the adaptor is plugged in.

The computer should be OFF when connecting the auxiliary power to either printer/plotter or computer. Also the two pronged plug should be connected to the wall outlet before connection or disconnection is made to the computer or printer/plotter.

About 15 hours will fully charge the rechargeable NiCad batteries in the printer/plotter; we find that during ordinary office use it is best to keep the unit plugged into the recharger or else an ERROR 80 (low battery) will show on the display and all printer/plotter/computer functions will stop.

If the printer/plotter is unplugged it will operate for about one eight hour day or less, depending on intensity of use of the machine. We expect this period of "unplugged" use to become shorter with time until the time when the NiCad batteries eventually must be replaced - - a job Radio Shack reserves to themselves.

We hope by keeping the printer/plotter typically "plugged in," to postpone the time when NiCad battery replacement is necessary. On the other hand, we seldom provide auxiliary power to the computer itself.

PAPER

The small rolls of 2¼" wide paper which are sold by Radio Shack, seem high

priced and are quickly exhausted. In following our natural frugal instincts, we have innovated a "bent-wire" spool holder for an ordinary roll of adding machine tape which is then fed through the back; in order to accomplish this, the cover must be taken off and then set loosely on top, secured only by the latch at the front. In this configuration, a slot exists in the rear through which the paper may be fed. We prefer the adding machine roll paper -- more "body", less "coil-spring" effect. However, if back drag develops or if the paper becomes hung up, it will slip in the rollers distorting the printout; this has happened to us a few times during plotting. So keep a careful watch on the paper.

We find that if the end of the tape is torn so as to make an "arrow-point" shape, the insertion of the tape into the rollers goes much more easily than if the squared-off end is inserted.

INSTALLING MEMORY MODULE

After installing a Memory Module it is necessary to type NEWO ENTER in the "PRO" mode, in order to merge the resident memory with the module memory. With the 8k module in place, a STATUS Ø should reveal available memory to be 1Ø,ØØ42 when memory has been cleared.

PC-2 AS A CALCULATOR

While in the "RUN" mode, a calculation expression may be entered through the keyboard (e.g., 8 * 5/2); then press ENTER , whereupon the answer, 20, is displayed. A continuation of the calculation can take the form of +6*5/10 ENTER whereupon the value, 23, is displayed. This calculation can be edited and rerun by use of the "cursor right" ( ) key which displays the previous calculation. The cursor can be moved to the location of the desired change; the numbers may be altered and a new answer obtained after pressing ENTER .

In order to save the result of a calculation, press cursor right, then SHIFT followed by cursor right, two times, and type A = followed by ENTER . The value is assigned to A and can be used in subsequent expressions.

For example: X=3*4/2, Y=8*9/24, Z=X/Y ENTER whereupon the answer, Z, is displayed; X, Y and Z are preserved in memory. Note the use of commas between expressions.

BEEP

The BEEP function is in the following form:

BEEP A, B, C
where
A= number of beeps (no default is permitted)
B= pitch of beep (default permitted)
C= duration of beep (default permitted)

After some experimentation we find that we prefer B to range between about 10 and 12 , and C between about 400 and 600. We do not like the "default" beep which is very high and very short. If the default beep is used, the number of beeps must, we feel, be at least four in order to be effective. We prefer:

A = 3
B = 11
C = 500

We suggest BEEP 3, 11, 500 (or a set of numbers meeting your preference) be assigned by a function key, to be used during programming.

RECOVERING INPUT AND OUTPUT DATA

While in the "Run" Mode, the values assigned to variables by the user, or calculated during execution of a program, may be recovered as follows: Type in the variable designation (e.g., A, X, RP, Q1, L$, NY(1), etc.), press ENTER ; the value or string stored in the computer memory, assigned to that variable, will be displayed. If the Printer-Plotter is in the "Echo" mode, the variable and the value or string, will be recorded on paper. Indeed, while in "Echo", the Printer-Plotter records all operations of the computer while it is being used as a calculator as above described, a very helpful feature.